

## 2.6 CAPTURA DE ENTRADA – “INPUT CAPTURE” CAPTURA DE EVENTOS EXTERNOS – MEDICIÓN DEL PERÍODO

Preparado por: Rangel Alvarado  
Estudiante Graduando de Lic. en Ing. Electromecánica  
Universidad Tecnológica de Panamá  
Panamá, Panamá  
“e-mail”: [issaiass@cwpanama.net](mailto:issaiass@cwpanama.net)  
“web site”: <http://www.geocities.com/issaiass/>

| ÍNDICE |  |
|--------|--|
| 2.6.1  | Introducción 248   |
| 2.6.2  | “Input Capture” 249  |
| 2.6.3  | Registros de Control de Configuración de Captura de Entrada - IC 250 |
| 2.6.4  | Diagrama de Flujo 251  |
| 2.6.5  | Código 253   |
| 2.6.6  | Simulación 258   |
| 2.6.7  | Conclusión 259   |
| 2.6.8  | Referencias 259  |
| 2.6.9  | Problemas Propuestos 259   |

### 2.6.1 Introducción

---

La última y poderosa opción de los canales del temporizador, captura de entrada (“input capture”) nos permite detectar el(los) flanco(s) de una señal. Con ella podemos detectar el tiempo en que una señal se inicia, su ciclo de trabajo y que tan frecuentemente ocurre.

Para esta nota se espera:

- Configurar una captura de entrada: inicializar una captura de entrada para reconocer una señal proveniente de alguna fuente externa.
- Simular: entender, por medio de “breakpoints”, cuando se genera una interrupción de captura de entrada y que tiempo transcurre luego de la acción del flanco.


*Nota: Se asume un dominio de la NT0101, NT0102, NT0103, NT0104 y se hace referencia cruzada a la misma si es necesario.*

## 2.6.2 “Input Capture”

La función “Input Capture” utiliza el contador libre de dieciseis (16) bits (TCNT[H:L]) para almacenar en el registro del canal (TCHx[H:L]) el correspondiente tiempo de un marco de referencia al detectarse un flanco activo (activo alto, bajo o ambos). Con este sistema se puede utilizar para:

- Tomar una referencia de tiempo de un evento externo
- Medir el período de la entrada
- Medir el ancho de un pulso de entrada
- Proveer interrupciones externas adicionales

*Si se desea medir el ancho del pulso (T1), se debe configurar la opción “input capture” para capturar flancos ascendentes y descendentes*



*Si se desea medir la frecuencia (T2) con que ocurre un evento, se debe de configurar la opción “input capture”, en este caso, para el flanco ascendente o para el ascendente.*

*Figura 146. Opción “Input Capture”. Dependiendo de los niveles lógicos de la onda, se debe configurar la opción “input capture” para tomar un flanco en específico si se captura la duración de un ciclo, ó, de ambos flancos si se desea medir el ancho de un pulso.*

## 2.6.3 Registros de Control de Configuración para Captura de Entrada - IC


### 2.6.3.1 Registros de Estado y Control del Canal del TIM - IC

Para el caso de captura de entrada solo se utilizan los bits MSxB, MSxA, ELSxB y ELSxA donde x es el número del canal (0 ó 1). Dependiendo de la configuración de estos bits, según la tabla 57 la captura es de flanco ascendente, descendente o ambos tipos de flancos.

Tabla 57. Modo de Selección de "Input Capture"

| MSxB | MSxA | ELSxB | ELSxA | Detección     |
|------|------|-------|-------|---------------|
| 0    | 0    | 0     | 1     | Flanco Alto   |
| 0    | 0    | 1     | 0     | Flanco Bajo   |
| 0    | 0    | 1     | 1     | Ambos Flancos |

Bits de Selección de nivel del flanco, ver Tabla 57.



0 = Comparación de Salida "Buffered" desactivada.  
 1 = Comparación de Salida "Buffered" activada.

1 = Comparación de Salida "Unbuffered".  
 0 = Captura de Entrada.

Nota: Si ELSX[B:A] ≠ 0:0, la combinación MSXA es válida.  
 MS1B no disponible, solo lee un cero.

Figura 147. Registro de Estado y Control del Canal del TIM – IC. Configurando los bits MSxB, MSxA a cero (0) y los bits ELSxB:A se pueden detectar por medio de capturas, la frecuencia, ancho de pulso y control de referencia de un evento externo.

### 2.6.3.2 Registros del Canal del TIM - IC

En la opción "input capture", contienen el valor de la cuenta del temporizador al detectarse el flanco.



Figura 148. Registros del Canal del TIM – IC. Al configurar el canal como "input capture" y al darse el flanco de captura, el TIM almacena en el registro del canal correspondiente el valor de la cuenta. (TCHx[H:L] ← TCNT[H:L]).

**2.6.3.3. Configuración de “Input Capture”**

Configure solamente el Canal 0 para detectar un flanco de bajada y habilite interrupciones.

Dirección: \$0025 - TSC0

|           |      |       |      |      |       |       |      |            |
|-----------|------|-------|------|------|-------|-------|------|------------|
| Leer:     | CH0F | CH0IE | MS0B | MS0A | ELS0B | ELS0A | TOV0 | CH0<br>MAX |
| Escribir: | 0    |       |      |      |       |       |      |            |
|           | 0    | 1     | 0    | 0    | 1     | 0     | 0    | 0          |

Figura 149. Configuración del Canal 0 del TIM – IC. El canal 0 es configurado para detectar un flanco de bajada y habilita las interrupciones del canal.

**2.6.4 Diagrama de Flujo**

El siguiente programa arroja en el registro H:X el resultado de la captura de la entrada, y retiene el periodo de la señal.




Figura 150. NT0106 – “Timer” – “Input Capture”. (a) Programa Principal. Prende y apaga el LED conectado al PTA5. (b) Reinicio del sistema. El sistema reinicia en cualquier circunstancia que se oprima el botón “reset”.




Figura 151. NT0106 – “Timer” – “Input Capture”. (a) Inicialización de Puertos. Inicializa el PORTA5 en Alto y como salida . (b) Inicialización del TIM. Inicializa el sistema para capturar el flanco de bajada y medir el período de una onda. (c) Interrupción del Canal. Cada vez que se capture un flanco, se reinicia el temporizador, teniendo el período aproximado de la onda. (d) Retardo. Subrutina de retardo programable de “hardware”.

## 2.6.5 Código

```
=====
; ARCHIVO      : NT0106 - Timer - Input Capture - 07 07 04.asm
; PROPÓSITO   : Retiene el valor de cada transición baja que este en el pin
;              : PTD4.
;
; NOTA        :
;              : 1 - Observar el resultado de la captura en H:X
;              : 2 - Observar el resultado del registro TCH0[H:L]
; REFERENCIA:
;              : Advanced Information of MCU68HC908JK1, JK3, JL3...
;              : http://www.freescale.com/files/microcontrollers/
;              : doc/data\_sheet/MC68HC08JL3.pdf
;              : Pág. 113 -Función de una Captura de Entrada, Sección 10.5.2
;              : Págs. 117 - 121. - Información de registros de Control
;              : del Canal
;              : Pág. 124 - Tabla de selección de modos del temporizador,
;              : Tabla 10-3.
;
;
; LENGUAJE    : IN-LINE ASSEMBLER
;
;-----
; HISTORY
; DD MM AA
; 07 03 03 Creado.
; 09 09 04 Modificado.
;
;-----
;Pasos para iniciar el TIM:
;
;              : 1 - Definir del registro TSC o registro de
;              : control del TIM:
;              : ¿Necesito o no interrupciones? (TOIE)
;              : ¿Cuál que valor de divisor ajustar? (PS[2:0])
;              : 2 - Definir el registro TMOD[H:L] como módulo máximo de conteo
;              :  $TMOD[H:L] = t*fx_{tal} / (2^{(2+PS)})$ 
;              : 3 - Habilitar Interrupciones globales (cli)
;              : 4 - Habilitar el temporizador (TSTOP = 0)
;              : 5 - Configurar el vector interrupción ($FFF2)
;              : 6 - Escribir el código de la interrupción y su retorno (rti)
;
;
;Pasos para iniciar el Canal del TIM
;              : 1 - Definir del registro TSCX registro de
;              : control del canal del TIM:
;              : ¿Necesito o no interrupciones? (CHXIE)
;              : ¿Input Capture, Output Compare, PWM?
;              : ¿Inicio en bajo o en alto? (MSXB,MSXA,ELSXB,ELSXA)
;              : 2 - Definir el registro TCHX[H:L] como módulo de comparación
;              :  $TCHX[H:L] = D * TMOD[H:L]$ , D = valor en porcentaje
;              : 3 - Configurar el vector interrupción ($FFF4 - CH1; $FFF6 - CH0)
;              : 4 - Escribir el código de la interrupción y su retorno (rti)
;
;=====
```

```

;=====
;
;                               Definiciones del Usuario
;=====
;
COPD      equ 0T                ; Bit 0 del registro CONFIG1
NULL      equ 0T                ; Valor nulo
MS1       equ 1T                ; retardo de 1 ms
PORTA5    equ 5T                ; PORTA, Bit 5 del Puerto A
BIT5      equ 5T                ; TSC, Bit de Inicio de Conteo, Bit 5 ON
DELAY49152 equ $0096           ; Constante de Retardo a 4.9152 MHz
CH0F      equ 7T                ; TSC0, Bit de Salida Comparada
TRST      equ 4T                ; TSC, Bit de Reinicio de conteo, Bit 4 ON
TSTOP     equ %00100000        ; TSC, Bit de pausa, Bit 5 ON
CH0IE     equ %01000000        ; Habilito interrupciones, canal 1
ELS0B     equ %00001000        ; Bit de Selección de flanco, activa...
;                               ; la señal en uno lógico al comparar

;=====
;
;                               Mapa de Memoria del Microcontrolador
;=====
;
;                               Registro de E/S
;=====
;
PORTA      equ $0000            ; Registro del Puerto A
DDRA       equ $0004            ; Registro de Direccionamiento del Puerto A

;=====
;
;                               Registro de Configuraciones
;=====
;
CONFIG1    equ $001F            ; Vectores de configuración

;=====
;
;                               Registro de Temporizador
;=====
;
TSC        equ $0020            ; Dirección, registro de estado y control del
;                               ; TIM
TSC0       equ $0025            ; Dirección, registro de estado y control del
;                               ; TIM Canal 0
TCH0H      equ $0026            ; TCH0H, Registro almacenador de
;                               ; comparación de la salida, registro alto.
TCH0L      equ $0027            ; TCH0L, Registro almacenador de
;                               ; comparación de la salida, registro bajo.

;=====
;
;                               Memoria FLASH
;=====
;
FLASH_START equ $EC00          ; Puntero - Mem.FLASH

```

```

;=====
;
;                               Vectores de Usuario
;=====
TIMCH0H    equ $FFF6            ; Vector de Sobreflujo del TIM, Canal 0(Alto)
RESET_VEC  equ $FFFE            ; Puntero del RESET

;=====
;
; OBJETIVO    : Inicio de Codif. del Ensam-
;              blador en Memoria FLASH.
;=====
                org FLASH_START            ; Inicio Mem. FLASH

;=====
;
; OBJETIVO    : Configura el TIM para cap-
;              turar un flanco de bajada y
;              así medir el período del
;              retardo.
;=====
START
    rsp                    ; Inic.Stack = $00ff
    bset COPD,CONFIG1     ; Desactiva watchdog
    clra                  ; Borra A
    jsr INIT_PORTS        ; Inicializa Puertos
    jsr INIT_TIM          ; Inicializa TIM
    cli                   ; Habilita Interrupciones
    bclr BIT5,TSC         ; Inicia el temporizado (TSTOP = 0)
ESPERA
    bset PORTA5,PORTA     ; Envía 1 lógico al puerto
    ldhx #MS1             ; A retardar 1 ms
    jsr Delay             ; A retardar 1 ms
    bclr PORTA5,PORTA    ; Envía 0 lógico al puerto
    ldhx #MS1             ; A retardar 1 ms
    jsr Delay             ; retarda 1 ms
    bra ESPERA            ; Salta al modo de bajo consumo

;=====
;
; INIT_PORTS : Inicializa variables y regis-
;              tros.
;
; OBJETIVO   : Inicializa los registros de
;              direccionamiento.
;              PORTA5 = HIGH
;              PORTA5 = OUTPUT
;
; ENTRADA    : Ninguna
; SALIDA     : Ninguna
; REGISTROS  :
; AFECTADOS  : PORTA, DDRA
;=====
INIT_PORTS
    bset PORTA5,PORTA     ; Fija PA5 = Alto
    bset PORTA5,DDRA     ; Fija PA5 = Salida
    rts                   ; retorna
    
```



```

=====
; INIT_TIM      : Inicializa el TIM
; OBJETIVO     : Inicialización del temporizador.
;              :
;              : Captura de Entrada por canal
;              : 0, interrupciones activadas
; ENTRADA      : Ninguna
; SALIDA       : Ninguna
; REGISTROS
; AFECTADOS    : TSC, TCH0H, TCH0L, TSC0, HX
=====
INIT_TIM
    ldhx #NULL                ; HX = 0
    sthx TCH0H                ; TCH0 = 0
    mov #{CH0IE|ELS0B},TSC0   ; Habilito Interrupciones del canal y configuro
                                ; una captura de entrada.
    rts                       ; retorna

=====
; DELAY        : Genera un retardo de tiempo
; OBJETIVO     : Retardo de tiempo, base 1ms
; ENTRADA      : H:X = Retardo en ms
; SALIDA       : H:X = 0
; REGISTROS
; AFECTADOS    : H:X
; USO          :
;              : MIN = H:X = 1T
;              : MÁX = H:X = 65535T
;              : ldhx #500
;              : jsr Delay ; retarda 0.5 seg
=====
Delay
    pshx                ; [2] Salva X en la pila
    pshh                ; [2] Salva H en la pila
    ldhx #DELAY49152    ; [3] Carga constante de bucle fino

Delay0
    aix #-1             ; [2] Decrementa H:X en 1
    cphx #0             ; [3] Llegó a cero (0)
    bne Delay0          ; [3] Si no es igual, salta a Delay0
    pulh                ; [2] Si es igual, recupera H de la pila
    pulx                ; [2] Recupera X de la pila
    aix #-1             ; [2] Decrementa H:X en 1
    cphx #0             ; [3] Llegó a cero (0)
    bne Delay           ; [3] Si no es igual, salta a Delay
    rts                 ; [4] retorna


```

```
=====
;
; TIMCH0L      : Interrupción del Canal 0 del
;               TIM
;
; OBJETIVO     : Reconoce la captura, flanco
;               de bajada
;
; ENTRADA      : Ninguna
; SALIDA       : Ninguna
; REGISTROS    :
; AFECTADOS    : TSC, TSC0, TCH0H, TCH0L,
;               H, STACK
;
=====
TIMCH0L
    pshh                ; Empuja H a la pila
    bclr CH0F,TSC0      ; Reconoce la interrupción del Canal 0
    bset TRST,TSC       ; reinicia el temporizador
    ldhx TCH0H          ; Carga con el período aproximado
                        ; en el segundo período.
    pulh                ; recuper H de la pila
    rti                 ; Retorna de la interrupción

=====
;
; OBJETIVO     : Inicializa el Vector de Reset
;               Arranque del programa en la
;               memoria Flash y búsqueda de
;               la interrupción del canal de
;               temporizado.
;
=====
;===== Vectores del TIM, captura de entrada =====
    org TIMCH0H          ; Inicio de los vectores del TIM
    dw TIMCH0L          ; Captura de flancos de bajada
;===== Vector de Reinicio de Sistema =====
    org RESET_VEC       ; Puntero Vec - RESET
    dw START            ; al darse reset salta a Start
```

*Listado 15. NT0106 – “Timer” – “Input Capture”. El programa enciende y apaga por medio de un retardo de software el PORTA5, el cual va cortocircuitado con el pin PORTD4 que es configurado como captura de entrada de flanco de bajada. El valor de la captura del período (aproximado) se puede visualizar en la segunda vez de captura, en el registro H:X.*

## 2.6.6 Simulación



(a) Inicie WinIDE.

(b) Cargue el archivo NT0106 – Timer – Input Capture – 09 07 04.asm.

(c) Compile.

(d) Entre al simulador.

(e) Añada un (1) "breakpoint" según la figura 152(a).

(f) Corra su programa paso a paso y en la sección de llamado a la rutina de inicialización del temporizador (INIT\_TIM) observe la configuración de captura de entrada, ver tabla 57.

(g) Presione el botón "Go", y espere a la llegada del "breakpoint". Nuevamente presione "Go", hasta llegar a la rutina de interrupción.

(h) Observe que la captura se ubica en el Registro del Canal 0 o también el programador depositó su valor en el registro H:X.

(j) Esta nota no está hecha para programación, no es un ejercicio de aplicación en el microcontrolador.

(d)

$$TCH0[H:L] = \frac{t \cdot f_{xial}}{2^{2+PS}}$$

$$09B4_{16} = \frac{t \cdot 4.9152 \times 10^6}{2^{2+0}}$$

$$t = 2.02ms$$

(4) Según la ecuación 5 y despejando, tenemos un período aproximado de 2.02 ms, concluyendo, el retardo de software no es precisamente 1 ms entre encendido y 1 ms entre apagado.

Figura 152. Simulación del Programa Principal. (a) "Breakpoint". "Breakpoints" para entender el mecanismo de captura de entrada. (b) Captura de Entrada. Luego de la segunda vez que su programa interrumpe, el canal 0, comandado por TCH0 registra el valor de tiempo de captura. (c) Captura en CPU. En este caso, el programador hizo una lectura para que también puede verse la captura en el registro H:X. (d) Tiempo transcurrido. Según notas anteriores, el tiempo transcurrido según las ecuaciones 5 y 6 es de 2.02 ms, corroborando que un retardo de "software" no es tan exacto como asumen algunos programadores.

## 2.6.7 Conclusión

---

La captura de entrada es un método efectivo para mediciones de ancho de pulso o períodos. Un ejemplo del buen uso de captura de entrada en una “aplicación seria” es el detectar el período de una onda a.c. para controlar el ángulo de disparo de una carga resistiva (“dimmer” digital).

Mediante el uso de “breakpoints” ud. corroboró que una captura de entrada por los canales del temporizador se puede efectuar y medir tiempos entre el rango del período o mayores (para medir períodos mayores ver referencia 2.6.8.1).

## 2.6.8 Referencias

---

### 2.6.8.1 Consideraciones del TOF al Medir Largos Períodos de Captura

(a) [http://www.freescale.com/files/microcontrollers/doc/eng\\_bulletin/EB389.pdf](http://www.freescale.com/files/microcontrollers/doc/eng_bulletin/EB389.pdf)

### 2.6.8.2 Información Avanzada sobre el Microcontrolador

(a) [http://www.freescale.com/files/microcontrollers/doc/data\\_sheet/MC68HC08JL3.pdf](http://www.freescale.com/files/microcontrollers/doc/data_sheet/MC68HC08JL3.pdf)

*Pág. 113 -Función de una Captura de Entrada, Sección 10.5.2*

*Págs. 117 - 121. - Información de registros de Control del Canal*

*Pág. 124 - Tabla de selección de modos del temporizador, Tabla 10-3.*

### 2.6.8.3 Manual de Referencia del CPU

(a) [http://www.freescale.com/files/microcontrollers/doc/ref\\_manual/CPU08RM.pdf](http://www.freescale.com/files/microcontrollers/doc/ref_manual/CPU08RM.pdf)

### 2.6.8.4 Página “web” sobre esta Nota Técnica

(a) <http://www.geocities.com/issaiass/>

## 2.6.9 Problemas Propuestos

---

2.6.9.1 Genere una captura de entrada por el Canal 1, de flancos de subida y detecte el cambio de un interruptor a uno (1) lógico.

2.6.9.2 Capture una señal de subida por el Canal 0 y ambos flancos en el Canal 1, cada canal con un interruptor diferente y deposite cuantas veces fue presionado en dos variables diferentes.